
Szanowni Państwo ! 
 
Oddajemy do rąk Państwa kolejną wersję naszego dokumentu opisującego zagadnienia związane z 
korzystaniem z oprogramowania SAS w polskich warunkach czyli aspekty lokalizacyjne SAS-a. 
Niniejsza jego wersja jest kontynuacją poprzednich dokumentów, przygotowanych od wersji 8, ale 
zawartością różni się od poprzednich – podobnie jak różni się SAS 9.2 od SAS-ów 9.1 i 8.2.  
Występujące w tytule określenie „SAS Foundation” oznacza te elementy SAS-a 9, które zostały 
odziedziczone po SAS-ie 8 i rozwinięte drogą ewolucyjną. Foundation jest fundamentem SAS-a 9.2.. 
Migracja SAS 8 lub SAS 9.1 Foundation do SAS 9.2 Foundation (tzw. „1 w 1”) jest względnie łatwa i 
nie powinna powodować większych problemów również w dziedzinie NLS-u (National Language 
Support), do czego m.in. ten dokument ma się przyczynić.  
W zasadzie wystarczy upewnić się, że opcja locale jest ustawiona na ‘Polish_Poland’, a SAS 
naturalnie dostosowuje się do specyfiki naszego języka i kraju. Niniejszy dokument wskazuje na 
napotkane przez nas na etapie testów odstępstwa od tej zasady, wymagające takiej czy innej uwagi. 
W wybranych sytuacjach umieściliśmy opis poprawnego działania elementów oprogramowania, aby 
zwrócić Państwa uwagę na wybraną funkcjonalność.  
W stosunku do poprzedniej edycji dokumentu powstał dodatkowy rozdział „SAS Accessy i zewnętrzne 
źródła danych”. 
Przygotowując tę edycję dokumentu testowaliśmy w polskim biurze SAS Institute wersję instalacji 
SAS-a 9.2 (TS2M3) na Windowsach.  
 „Polonizacja Foundation SAS-a 9.2 platformie Windows” powinna zostać udostępniona każdej osobie 
stykającej się z zagadnieniami związanymi z polskimi znakami w tworzonych aplikacjach. Lektura 
może oszczędzić wysiłek w czasie tworzenia i uruchamiania aplikacji. 
 
Jednocześnie zaznaczamy, że dokument ten jest uzupełnieniem, a nie zastąpieniem dokumentacji. W 
szczególności polecamy publikację SAS 9.2 National Language Support (NLS) Reference Guide 
dostępną dla wszystkich naszych Klientów w SAS-owym helpie lub OnlineDocu. 
 
Dokument ten i inne informacje związane z tematem będą publikowane na naszych stronach WWW 
pod adresem: 

http://www.sas.com/poland/services 
 

W razie dodatkowych pytań prosimy o kontakt z naszym Zespołem, telefonicznie na numer 
(22) 5604666 lub elektronicznie pod adresem support@spl.sas.com.  
 
Prosimy o uwagi i spostrzeżenia, które pomogą nam w przygotowaniu następnych wersji tego 
dokumentu.  
 

Zawsze służąc Państwu pomocą, 
Zespół Pomocy Technicznej Centrum Technologii  SAS Polska 

1. BASE 
 
1.0.  Polska instalacja SAS-a 
 
SAS Deployment Wizard za pomocą którego wykonujemy instalację jest dostępny w wersji angielskiej. 
 
W trakcie instalacji (język instalacji: Angielski) pojawia się pytanie o wybranie dodatkowych języków. 
Zaznaczany Polski, jeżeli chcemy mieć możliwość korzystania z polskiego interfejsu. Polska wersja 
SAS-a ma polskie menu, polskie okienka oraz w znacznym zakresie polskie komunikaty.  
 
Uruchomienie polskiej wersji SAS-a, standardowo ikoną „SAS 9.2 (Polish)”, nie jest konieczne, aby 
korzystać z udogodnień przy przetwarzaniu z językiem polskim (polskie znaki, daty, itd.). Niezbędne 
jest tylko ustawienie opcji locale.  
Na polskich Windowsach (istotne są polskie ustawienia regionalne) opcja locale jest podczas instalacji 
automatycznie ustawiona na " PL_PL" (co jest równoważne ustawieniom Polish_Poland lub Polish). 
Dzieje się tak również dla angielskiej wersji SAS-a 9.1 (uruchamianej ikoną „SAS 9.1 (English)”). 
Ustawienie tej opcji znajduje się w sas’owym configu (dla polskiej instalacji w <folder instalacyjny 
SASa>\SASFoundation\9.2\nls\pl\SASV9.CFG). 
 
Opcja locale odzwierciedla lokalne konwencje języka, kraju lub regionu. Ustawia polskie czcionki, 
formę przedstawiania daty, czasu, liczb i waluty. 
Wraz z ustawieniem opcji locale, aby ułatwić użytkownikom pracę z językiem narodowym, 
automatycznie ustawiane są opcje: encoding, datestyle, dflang, datestyle, papersize. 
 
1    proc options group=languagecontrol; 

2    run; 

 

    SAS (r) Program chroniony prawem Edycja 9.2  TS2B0 

 

 DATESTYLE=YMD     Identyfikuje sekwencje miesiąca, dnia i roku z niejednoznacznym informatem 

ANYDATE 

 DFLANG=POLISH     Język europejskich formatów i informatów daty i godziny 

 NOLOCALELANGCHG   Bez zmiany języka tekstu komunikatów w wynikach przy podanej opcji LOCALE 

 PAPERSIZE=LETTER  Format papieru do drukowania 

 RSASIOTRANSERROR  Wyświetla błąd transkodowania przy próbie zdalnego zastosowania 

niedozwolonych wartości danych 

 TRANTAB=(lat1lat1,lat1lat1,wlt2_ucs,wlt2_lcs,wlt2_ccl,,,) 

                   Nazwy tablic translacji 

 NODBCS            Nie przetwarza zestawów znaków dwubajtowych 

 DBCSLANG=NONE     Określa używany język o dwubajtowym zestawie znaków (DBCS) 

 DBCSTYPE=NONE     Określa metodę kodowania dwubajtowego zestawu znaków (DBCS) 

 ENCODING=WLATIN2  Określa domyślne kodowanie dla wewnętrznego przetwarzania danych. 

 LOCALE=POLISH_POLAND 

                   Określa bieżące ustawienia lokalne sesji SAS-a. 

 NONLSCOMPATMODE   Używa do przetwarzania danych znakowych kodowania określonego przez 

użytkownika 

 
Opcja encoding jest jedyną z powyższej listy, której wartość nie może być zmieniona w trakcie sesji. 
 
Pomimo ustawienia opcji papersize na letter do drukowania okienek sasowych jest wybierany 
poprawny format A4.  
 
 
Oprócz opcji locale szczególną rolę pełni opcja encoding. Ustawienie tej opcji decyduje o interpretacji 
kodów polskich liter w czasie przetwarzania. Każdy zbiór w formacje SAS 9 posiada w nagłówku 
informacje o stronie kodowej oraz o platformie systemu operacyjnego na jakiej został zapisany.  
 
Oto przykład nagłówka zbioru: 
                                              System SAS            15:51 Wednesday, April 2, 

2008   1 

 

                                          Procedura CONTENTS 

 

       Nazwa zbioru            WORK.AIR                             Obserwacje             144 

       Typ elementu            DATA                                 Zmienne                2 

       Mechanizm               V9                                   Indeksy                0 

       Utworzono               środa, 02 kwiecień 2008 15:58:16     Długość obserwacji     16 

       Ostatnio zmodyfikowano  środa, 02 kwiecień 2008 15:58:16     Usuniętych obserwacji  0 

       Zabezpieczenie                                               Skompresowany          Nie 

       Typ zbioru                                                   Posortowany            Nie 

       Etykieta 

       Reprezentacja danych    WINDOWS_32 

       Kodowanie               wlatin2  Central Europe (Windows) 

 

 

                        Informacje zależne od mechanizmu/systemu operacyjnego 

 

Rozmiar strony zbioru              4096 

Liczba stron zbioru                1 

Pierwsza strona                    1 

Maksymalnie obserwacji na stronie  252 

Obserwacji na pierwszej stronie    144 

Liczba napraw zbioru               0 

Nazwa pliku                        C:\DOCUME~1\sasadm\USTAWI~1\Temp\SAS 

                                   Temporary Files\_TD2992\air.sas7bdat 

Utworzono w edycji                 9.0201M0 

Utworzono w systemie op.           XP_PRO 

 

 

                               Alfabetyczna lista zmiennych i atrybutów 

 

          #    Zmienna    Typ     Dł.    Format    Etykieta 

 

          2    AIR        Num       8              international airline travel (thousands) 

          1    DATE       Num       8    MONYY.    

 

 
Ustawienie opcji encoding (na poziomie sesji, libname, filename, pojedynczego zbioru) nie tylko 
decyduje o stronie kodowej nowo utworzonego zbioru czy pliku zewnętrznego, ale także może 
decydować o interpretacji odczytywanych danych (tekstowych) i ewentualnym przekodowaniu 
czytanych zbiorów (jeżeli mają inne kodowanie niż kodowanie sesji).  
Automatyczne przekodowania danych między różnymi encodingami (transcoding) wykonuje 
mechanizm CEDA (Cross Environment Data Exchange).  
Dodatkowo można też ustawić opcje biblioteki inencoding i outencoding wymuszając encoding 
zbiorów czytanych i zapisywanych do niej zbiorów. 
W przypadku częstej konieczności pracy z różnymi encodingami zachęcamy do lektury dokumentacji 
SAS-a, w szczególności rozdziału: SAS Products-> SAS 9.2 National Language Support Reference 
Guide-> NLS Concepts-> Encoding for NLS. 
 
Encoding można też ustawić tak, aby obsługiwał jeden ze standardów Unicode. Uwagi na temat 
używania Unicodu znajdują się w rozdziale 1.8. 
 
Uwagi o ustawianiu opcji encoding dla sasowych serwerów w architekturze SAS-a 9 znajdują się w 
rozdziale 5. 
 
 
1.1. DMS – Display Manager System 
 
Nie powinno brakować polskich czcionek ani w menu sasowym (po polsku w polskiej instalacji SAS-
a), ani w okienkach sasowych. Gdyby u Państwa ich nie było należy sprawdzić, czy wybrana czcionka 
ma ustawiony skrypt ‘Środkowoeuropejski’ albo ‘Europa Środkowa’. Czcionkę w Enhanced Edytorze 
ustawiamy w Narzędzia-> Opcje-> Enhanced Editor-> Wygląd. Czcionkę w okienkach Edytor 
programów, Log i Wynik ustawiamy przez Narzędzia-> Opcje-> Czcionki… przy zaznaczonym jednym 
z tych okienek.  
 
Jeżeli zmienimy czcionkę to możemy takie ustawienia zapamiętać dla następnych sesji SAS-a, jeżeli 
zaznaczymy pole „Zapisuj ustawienia końcowe” w Narzędzia-> Opcje-> Preferencje-> Ogólne. 
 
1.1.1. Explorer (SAS-owy) 
 
Sortowanie Windows-owych folderów lub plików (w Exploratorze przez Mój Komputer lub Ulubione 
foldery i przy ustawieniu Widok-> Szczegóły) w wersji 9.2 uwzględnia polskie znaki. 

1.2 Polskie sortowanie. 
 
Aby sortować po polsku wystarczy ustawić opcję ‘sortseq’. 
Najprościej jest ustawić ją domyślnie dla wszystkich użytkowników w sasv9.cfg (<folder instalacyjny 
SASa>\nls\pl\sasv9.cfg): 
 
-sortseq Polish 

 
Dla testu można wykonać: 
 
* jeżeli opcja sortseq jest ustawiona globalnie to poniższa linia jest zbędna; 

options sortseq=Polish; 

 

data t; 

  length dane $20; 

  input dane @@; 

  cards; 

zbir żuk Monika ą źdźbło maszyna Lew łopata Ą Ala analiza lwica Łódź Adam 

; 

run; 

proc sort data=t; 

by dane; 

title1 'Zbiór po posortowaniu słownikowym'; 

proc print; 

run; 

 

Procedura SORT, po ustawieniu opcji sortseq (=Polish) będzie wykonywała polskie sortowanie w 
takim porządku: aAąĄbBcCćĆ. Jest to inny porządek sortowania niż (domyślny przy ustawionej opcji 
sortseq) w SAS 9.1.3: AĄBCĆ...aąbcć.  
W  przypadku potrzeby sortowania w innym porządku komputerowym, udostępniamy makro, które 
stworzy potrzebną tablice sortowania.  
 
Nowością w 9.2 jest możliwość ustawienia sortowania słownikowego – opartego na zasadach 
sortowania języka (ustawionego w opcji locale) i przypominającego sortowanie jakie spotykamy w 
słownikach. Takie sortowanie ustawia się przez opcję procedury sort sortseq=LINGUISTIC.  
Przykładowo wartości „Kat kat kot” przy ustawionej opcji sortseq=Polish procedura sort ułoży w 
porządku: „Kat kat kot”, a  przy sortseq=LINGUISTIC w porządku: „kat Kat kot”. Reguły sortowania 
słownikowego można zmodyfikować za licznych pomocą opcji. Szczegóły na temat „LINGUISTIC 
storting”:  
http://support.sas.com/resources/papers/linguistic_collation.pdf 
Jedna z opcji pozwala tak posortować tekst tak, aby uwzględniał znajdujące się w nim liczby np. 
numery domów w adresach (aby "Informatyczna 201" został umieszczony po "Informatyczna 21"). Aby 
to osiągnąć w 9.2 wystarczy zbiór posortować z takimi opcjami: 
proc sort sortseq=linguistic(numeric_collation=on); 
 
Jeżeli ustawimy opcję sortseq globalnie możemy napotkać na błąd: 
ERROR: Input data sets cannot be combined because they have different  

collating sequences (SORTSEQ). 

Nota : 
http://support.sas.com/kb/35/644.html 
opisuje jego przyczyny i proponuje obejście. 
 
 
1.3 Polskie formaty i funkcje: liczbowe, daty, godziny, waluty. 
 
Ustawienie opcji locale na Polish (domyślne dla polskiej instalacji SASa) ułatwia używanie polskich 
formatów. 
 
 
 
 

http://support.sas.com/resources/papers/linguistic_collation.pdf
http://support.sas.com/kb/35/644.html

Przykłady przedstawienia liczby z przecinkiem dziesiętnym: 
 
data _null_; 

l=-3384.70; 

put l; 

put l numx12.2; *bez spacji między tysiącami; 

put l nlnum12.2; *ze spacją między tysiącami - przy locale=Polish; 

put l commax12.2;  

*długość formatów jest przykładowa ; 

run; 

 
Przykłady użycia explicite polskich formatów daty: 
 
data _null_; 

data='7oct1960'd; 

put data; 

put data date.; put 'i po polsku:'; 

put data poldfde.;  

put data poldfdt.;  

put data poldfdd.;  

put data poldfdwn.; 

put data poldfmn.; 

put data poldfmy.; 

put data poldfwkx.;   

put data poldfdn.; 

put data poldfwdx.; 

amount=1; 

put 'przeliczenie 1 euro na pln - bieżący kurs można skonfigurować:' amount 

eurtoplz10.2 ' zł';  

amount=10; 

put amount eurfrplz9.2; 

run; 

 
Te same efekty uzyskamy używając odpowiednich formatów NLxxxx lub EURxxxx. Ponieważ przy 
ustawionej opcji locale=Polish opcja dflang jest ustawiana również na Polish, formaty NLxxxx i 
EURxxxx. będą interpretowane jako polskie. 
W 9.2 pojawiły się też nowe formaty uwzględniające specyfikę naszego języka i kraju. Te dla 
wyróżnienia mają w poniższym przykładzie duże litery w nazwie. 
 

data _null_; 

data='7oct1960'd; 

put data nldate.; * date or worddate equivalent; 

put data nldatemn.; * monname equivalent; 

put data nldatew.; * weekdate equivalent; 

put data nldatewn.; * print day of week name; 

put data nldatm.; * datetime equivalent; 

put data nldatmtm.; * tod equivalent; 

put data nldatmw.; * twmdy equivalent; 

amount=123456.78; 

put amount nlmny15.2; 

put amount NLMNIPLN15.2; 

put amount NLMNLPLN15.2; 

put amount NLBEST.; 

put amount euro15.2;  

put amount nlmnieur15.2; * alternatywny sposób pokazania symbolu euro;  

run; 

 
Oprócz formatów można też używać szeregu informatów i funkcji uwzględniających polskie konwencje 
językowe. 
Analogicznymi do formatów nowymi informatami w 9.2 są: NLMNIPLNw.d, NLMNLPLNw.d. 
 
W przypadku używania formatów, których domyślne długości są niewystarczające dla wyników w 
języku polskim należy explicite podać długość formatu. W przeciwnym wypadku SAS może, np. 
zamiast pełnej nazwy wypisać mnemonik. Tę sytuację ilustruje następujący fragment logu. 

1    options locale=Polish_Poland; 

2 

3    data _null_; 

4    i=3; 

5    weekname=put(i,nldatewn12.); 

6    put weekname; 

7    weekname=put(i,nldatewn.); 

8    put weekname; 

9    run; 

 

poniedziałek 

Pn 

  
O formatach i informatach związanych z Unicodem wspominamy w rozdziale 1.8. 
 
 
1.4 ODS 
 
W rezultatach zdefiniowanych przez polecenie ods (typu np. html, rtf, pdf) widać polskie znaki. 
 
Nie ma potrzeby, jaka była z wersją 9.1.3, stosować się do noty SN-009650.  
 
 
1.4.1 PDF generowany z aplikacji klienckich (np. Enterprise Guide, SAS Olap Viewer) 
 
Poniższy paragraf i kod dotyczy Państwa, jeżeli nie mają Państwo wersji SAS-a 9.2 (TSxMy), gdzie 
x>=2 (czyli fazy 2. lub 3.).   
Nie każdy PDF wygenerowany przez SAS-a jest tworzony przy użyciu mechanizmu ODS. Część 
aplikacji klienckich wykorzystuje do tego celu mechanizm tzw. wirtualnych drukarek PDF. Ustawienia 
dotyczące tego mechanizmu zawarte są w rejestrze SAS-a. Niestety, domyślne ustawienia wirtualnej 
drukarki PDF nie zezwalają na utworzenie pliku PDF z poprawnie wyświetlanymi polskimi znakami. 
Potrzebna jest modyfikacja rejestru SAS-a. Poniższy kod realizuje przykładową modyfikację rejestru i 
pozwala na utworzenie poprawnych plików PDF w aplikacjach klienckich. 
 
/* Zarejestruj wszystkie czcionki systemowe w SAS-ie */ 

proc fontreg; 

 fontpath "%sysget(systemroot)\Fonts"; 

run; 

 

/* Wyeksportuj wszystkie czcionki zarejestrowane w bibliotece SASUSER do pliku 

   fonts.sasxreg */ 

proc registry export="D:\fonts.sasxreg" startat="core\printing\freetype\fonts";  

run; 

 

/* Importuj zawartość pliku fonts.sasxreg do biblioteki SASHELP */ 

proc registry import="D:\fonts.sasxreg" usesashelp; 

run; 

 

/* W pliku ttfpdf.sasxreg utwórz aliasy dla rodziny czcionek PDF */ 

data _null_; 

 file "D:\ttfpdf.sasxreg"; 

 input line $80.; 

 put line; 

 lines; 

[CORE\PRINTING\PDF\FONT FAMILIES] 

"AvantGardeGothic"="<ttf> MS Gothic" 

"Bookman"="<ttf> Times New Roman" 

"Chancery"="<ttf> Times New Roman" 

"Courier"="<ttf> Courier New" 

"Dingbats"="<ttf> Wingdings" 

"Helvetica"="<ttf> Arial" 

"NewCenturySchoolbook"="<ttf> Times New Roman" 

"Palatino"="<ttf> Palatino Linotype" 

"Symbol"="<ttf> Symbol" 

"Times"="<ttf> Times New Roman" 

[CORE\PRINTING\ALIAS\FONTS\PDF] 

"Century Schoolbook"="<ttf> Century Schoolbook" 

"NewCenturySchlbk"="<ttf> Times New Roman" 

"Times New Roman"="<ttf> Times New Roman" 

"ZapfChancery"="<ttf> Times New Roman" 

"ZapfDingbats"="<ttf> Wingdings" 

"ITC Bookman"="<ttf> Times New Roman" 

"Courier"="<ttf> Coutier New" 

"SAS Monospace"="<ttf> Courier" 

"Palatino"="<ttf> Palatino Linotype" 

"Times"="<ttf> Times New Roman" 

"monospace"="<ttf> Courier New" 

"san-serif"="<ttf> Arial" 

"serif"="<ttf> Times New Roman" 

[CORE\PRINTING\PRINTERS\PDF\ADVANCED] 

"FONT FAMILIES"=LINK:"\\CORE\\PRINTING\\PDF\\FONT FAMILIES" 

[CORE\PRINTING\PRINTERS\PDF\DEFAULT SETTINGS] 

"Font Typeface"="<ttf> Courier New" 

[CORE\PRINTING\PROTOTYPES\PDF Version 1.3 (Partial Font Embedding)\ADVANCED] 

"FONT FAMILIES"=LINK:"\\CORE\\PRINTING\\PDF\\FONT FAMILIES" 

[CORE\PRINTING\PROTOTYPES\PDF Version 1.3 (Full Font Embedding)\DEFAULT SETTINGS] 

"Font Typeface"="<ttf> Courier New" 

[CORE\PRINTING\PROTOTYPES\PDF Version 1.3 (Full Font Embedding)\ADVANCED] 

"FONT FAMILIES"=LINK:"\\CORE\\PRINTING\\PDF\\FONT FAMILIES" 

[CORE\PRINTING\PROTOTYPES\PDF Version 1.3\DEFAULT SETTINGS] 

"Font Typeface"="<ttf> Courier New" 

[CORE\PRINTING\PROTOTYPES\PDF Version 1.3\ADVANCED] 

"FONT FAMILIES"=LINK:"\\CORE\\PRINTING\\PDF\\FONT FAMILIES" 

[CORE\PRINTING\PROTOTYPES\PDF Version 1.2 (Partial Font Embedding)\DEFAULT SETTINGS] 

"Font Typeface"="<ttf> Courier New" 

[CORE\PRINTING\PROTOTYPES\PDF Version 1.2 (Partial Font Embedding)\ADVANCED] 

"FONT FAMILIES"=LINK:"\\CORE\\PRINTING\\PDF\\FONT FAMILIES" 

[CORE\PRINTING\PROTOTYPES\PDF Version 1.2 (Full Font Embedding)\DEFAULT SETTINGS] 

"Font Typeface"="<ttf> Courier New" 

[CORE\PRINTING\PROTOTYPES\PDF Version 1.2 (Full Font Embedding)\ADVANCED] 

"FONT FAMILIES"=LINK:"\\CORE\\PRINTING\\PDF\\FONT FAMILIES" 

[CORE\PRINTING\PROTOTYPES\PDF Version 1.2\DEFAULT SETTINGS] 

"Font Typeface"="<ttf> Courier New" 

[CORE\PRINTING\PROTOTYPES\PDF Version 1.2\ADVANCED] 

"FONT FAMILIES"=LINK:"\\CORE\\PRINTING\\PDF\\FONT FAMILIES" 

[CORE\PRINTING\PROTOTYPES\PDF Version 1.3 (Partial Font Embedding)\DEFAULT SETTINGS] 

"Font Typeface"="<ttf> Courier New" 

run; 

 

/* Importuj aliasy rodziny czcionek PDF do rejestru SAS-a */ 

proc registry import="D:\ttfpdf.sasxreg" usesashelp;  

run; 

 
 
1.5 Polskie drukowanie 
 
Na wydruku z okienek tekstowych polskie czcionki powinny pojawiać się bez żadnych ustawień. 
Jeżeliby się nie pojawiały, należy w ustawieniach drukarki Plik-> Drukuj-> Ustawienia-> Czcionka 
zmienić czcionkę na taką, która polskie znaki posiada, pamiętając o zaznaczeniu odpowiedniego 
skryptu. Nowa czcionka zostanie zapamiętana  jako domyślna dla drukowania. Alternatywnie można 
ustawić opcję sysprintfont. 
 
 
1.6 Polskie nazwy zmiennych 
 
Jeżeli chcemy, aby w nazwach zmiennych  występowały polskie znaki należy ustawić opcje 
validvarname=any.  
 
 
1.7 SPDE i „polskie ogonki” 
 
Motor SPDE w 9.2 prawidłowo wykonuje funkcje upcase/lowcase z polskimi znakami.  
SPDE nie wspiera podłączenia się przez mechanizm RLS do biblioteki z sesji na innej platformie 
operacyjnej (np. z Unixa). W związku z tym nie jest potrzebne (a także wspierane) przekodowanie 
(polskich) znaków.  

1.8 Unicode 
 
Dane w zbiorze mogą być też zapisane w jednym ze standardów Unicode (utf-8). Zastosowanie 
takiego rozwiązania można sobie wyobrazić w sytuacji, gdy chcemy w zmiennych tekstowych 
zapisywać słowa z języków o różnych encodingach (np. wlatin2 i wlatin1 – odpowiednio Grażyna i 
Jürgen).  
Stosując Unicode powinniśmy pamiętać m.in. o tym, że na Unicodzie 

 operują inne funkcje tekstowe – zaczynają się od ‘k’ 

 inne formaty – zaczynają się od ‘u’  

 zmienne tekstowe mogą zajmować więcej miejsca niż zadeklarowaliśmy, gdyż 1 litera może 
być opisana nie na 1B jak w przypadku wlantinx, ale może zajmować do 4B. 

 
Gdy dane są zapisane w Unicode, w jednej sesji SAS-a, bez kwestii encodingu (którego nie można w 
sesji zmieniać), można stosować różne wartości opcji locale. Jest to opisane w dokumencie: 
http://support.sas.com/resources/papers/LocaleSwitching.pdf 
 
W SAS-ie 9.2 ActiveX wspiera fonty Unicode.  
 
Do odczytu zbioru/zapisu pliku tekstowego w Unicode ustawienie opcji encoding na utf-8 nie jest 
konieczne. Wystarczy użyć opcji encoding w instrukcjach: filename lub (w data stepie) (in)file. 
Dostępne są wartości: utf-8, utf-16 i utf-32. 
  
Więcej informacji na temat przetwarzania wielojęzycznych danych w SAS-ie w artykule "Processing 
Multilingual Data with the SAS® 9.2 Unicode Server": 
http://support.sas.com/resources/papers/92unicodesrvr.pdf 
  
 
2. GRAPH 
 
2.1 Czcionki systemowe (HARDWARE FONTS). 
 
Jak w poprzednich wersjach, przy generowaniu grafiki, można korzystać z skorzystać z czcionek 
TrueType dostępnych w systemie operacyjnym Windows. Jeżeli korzystamy z polskiej wersji Windows 
wystarczy używać nazw czcionek jako wartości odpowiednich opcji. Dodając do nazwy czcionki /bo, 
uzyskamy pogrubienie, a /it - czcionkę pochyłą.  
Przykładowy program demonstrujący użycie czcionek: 
data klasa; 

  length wzrost 8 kategoria $ 10; 

  input  wzrost kategoria $; 

  kategoria_u=upcase(kategoria);  kategoria_l=lowcase(kategoria); 

  label kategoria_u='Kat. ąęćłńóśźż'   kategoria_l='Kat. ĄĘĆŁŃÓŚŹŻ'; 

cards; 

181     Liść 

167     Sęk 

180     Źdźbło 

181     Źdźbło 

172     Kość 

181     Dąb 

; 

run; 

goptions reset=all; 

title1 f='Courier/it' h=10 pct 'Kat. ąęćłńóśźż'; 

title2 f='Tahoma/bo' h=6 pct 'Kat. ĄĘĆŁŃÓŚŹŻ'; 

legend1 label=(font='Arial/bo/it' 'Legenda ĄĘĆŁŃÓŚŹŻ') 

        value=(font='Arial/it/' height=1.5 pct); 

axis1   label=(font='Times New Roman/bo/it' 'Axis ĄĘĆŁŃÓŚŹŻ');      

goptions ftext='Arial'; 

proc gchart data= klasa; 

  vbar kategoria_l /type=mean sumvar=wzrost  

        maxis=axis1 

        subgroup=kategoria_u legend=legend1; 

  run; 

quit; 

http://support.sas.com/resources/papers/LocaleSwitching.pdf
http://support.sas.com/resources/papers/92unicodesrvr.pdf

Uzyskana na ekranie grafika zostanie poprawnie wydrukowana lub zapisana w jednym z wielu 
formatów graficznych (m.in. GIF, JPEG, WMF).  
W identyczny sposób (zmieniając jedynie urządzenie docelowe) możemy wygenerować grafikę 
bezpośrednio na drukarkę lub do pliku zewnętrznego. 
 
 
2.2 Czcionki SAS/GRAPH (SOFTWARE FONTS). 
 
Niektóre czcionki SAS/Graph można wykorzystać do produkcji grafiki z polskimi literami niezależnie 
od systemu. Czcionki zawierające nasze „ogonki” to CENTX, SWISS, ZAPF. Te właśnie nazwy 
powinny zostać użyte w specyfikacjach czcionek. Dodatkowo należy ustawić odpowiednią tablicę 
przekodowań (keymapę) ze strony systemowej (Windows 1250) na wewnętrzną stronę kodową 
SAS/GRAPH. Tablicę przekodowań tworzymy wykonując poniższy kod, ale najpierw musimy 
dowiązać bibliotekę o nazwie gfont0, w której zostanie wygenerowana tablica przekodowań (kod na 
następnej stronie). 
 
/* generowanie keymapy dla WINDOWS */ 

libname gfont0 (sashelp); 

 

proc gkeymap name=winansi out=temp keymap; 

run; 

data mod; 

    retain tolen 1; 

    from="d3"X;TO="08"x;output;  /* Oacute   */ 

    from="f3"X;TO="a2"x;output;  /* oacute   */ 

    from="f1"X;TO="c0"x;output;  /* nacute   */ 

    from="d1"X;TO="c1"x;output;  /* Nacute   */ 

    from="9c"X;TO="c2"x;output;  /* sacute   */ 

    from="8c"X;TO="c3"x;output;  /* Sacute   */ 

    from="9f"X;TO="c4"x;output;  /* zacute   */ 

    from="8f"X;TO="c5"x;output;  /* Zacute   */ 

    from="bf"X;TO="c6"x;output;  /* zdot     */ 

    from="af"X;TO="c7"x;output;  /* Zdot     */ 

    from="b3"X;TO="c8"x;output;  /* lslash   */ 

    from="a3"X;TO="c9"x;output;  /* Lslash   */ 

    from="b9"X;TO="ca"x;output;  /* ahook   */ 

    from="a5"X;TO="cb"x;output;  /* Ahook   */ 

    from="ea"X;TO="cc"x;output;  /* ehook   */ 

    from="ca"X;TO="cd"x;output;  /* Ehook   */ 

    from="c6"X;TO="f4"x;output;  /* Cacute  */ 

    from="e6"X;TO="f9"x;output;  /* cacute  */ 

run; 

proc sort data=mod; by from; 

run; 

data temp; 

  update temp mod; by from; 

run; 

proc gkeymap name=win1250 data=temp keymap; 

run; 

 
A oto przykładowy kod korzystający z polskich znaków: 
 
goptions reset=all; 

libname gfont0 (sashelp); 

goptions keymap=win1250; 

proc gslide; 

  note f=swiss  h=3 'Żołądek'; 

  run; 

quit; 

 
Alternatywnie keymapy i devmapy można stworzyć makrem %lsgraph(wlt2); 
 
Informacje nt. tworzenia keymapy i devmapy znajdują się również w odpowiednim dla platformy 
Configuration Guide for SAS® 9.2 Foundation. 

2.3 Mapa Polski. 
 
Podstawowy zbiór z mapą - Poland przedstawia podział na powiaty. Dostępny jest również zbiór 
Poland2, który zawiera powiązania powiatów z województwami. Aby otrzymać mapę z 
województwami, należy odpowiednio te dwa zbiory połączyć. Następnie użyć procedury GREMOVE, 
aby usunąć granice powiatów. Przedstawia to poniższy kod: 
 

data poland; 

merge maps.poland maps.poland2; 

by id; 

run; 

proc sort data=poland; 

by wojid id; 

run; 

proc gremove data=poland out=poland;  

id id;  

by wojid;  

run; 

 
Aby zobaczyć mapę z województwami w SAS-ie można np.: 
 
data dane; 

do i=1 to 16; 

  wojid=i; value=i;  

  output; 

end; 

drop i; 

run; 

 

goptions reset=all; 

proc gmap data=dane map=poland; 

id wojid; 

choro value/levels=16; 

run; 

quit; 

 
 
2.4 Polskie sortowanie na wykresach SAS/GRAPH . 
 
Wartości (tekstowe) na osiach wykresów (np. z procedury gchart) są sortowane po polsku.  
 
 
3. CONNECT i SHARE 
 

O ile tylko sesja SAS-a odbierająca dane (np. Connect Server w przypadku proc upload) będzie miała 
ustawioną opcję locale na Polish, nie powinni Państwo napotkać w tych obszarach problemów.  
W przypadku komunikacji SAS 9 – SAS 9 z ewentualną translacją poradzi sobie mechanizm CEDA. W 
przypadku komunikacji SAS 9 – SAS 8 odpowiednie tablice zostaną automatycznie udostępnione. 
 

 
4. Ustawienie opcji encoding dla serwerów w architekturze SAS 9. 
 
Serwery w architekturze SAS9 działają z kodowaniem (w encoding’u), który jest ustawiany w aplikacji 
klienckiej (np. DI Studio możliwe są ustawienia  
JavaArgs_13=-Duser.language=pl 
JavaArgs_14=-Duser.country=pl)  
albo dla serwera/serwerów, np. w pliku sasv9_usermods.cfg. 
Nota http://support.sas.com/kb/35/644.html wskazuje, że ustawienie dla serwera są późniejsze więc 
efektywne. 
Sprawdzenie ustawienia opcji encoding dla sesji SAS-a (Workspace, Stored Process) wykonuje się 
tak samo jak dla zwykłej sesji SAS (proc options option=encoding). Jeżeli ustawienie opcji nie jest 
zgodne z oczekiwaniami, przyczynę pojawienia się innej wartości można szukać uruchamiając proc 

http://support.sas.com/kb/35/644.html

options option=config, i weryfikując w kolejnych configach (zaczynając od ostatniego), gdzie ta 
wartość jest ustawiona lub sprawdzając ustawienia regionalne/kulturowe aplikacji klienckiej. 
 
 
5. SAS Accessy i zewnętrzne źródła danych 
 
W przypadku problemów z widocznością polskich znaków pochodzących z zewnętrznego (dla SAS-a) 
źródła danych (np. Excel, MS SQL Server) proponujemy zweryfikować ustawienia zgodnie z notą 
(aktualną dla wersji 9.2, oprócz linka do dokumentacji): 
http://support.sas.com/kb/18/849.html 
ze szczególnym uwzględnieniem punktu 1d, który wskazuje na częstą przyczynę problemów – 
nieustawienie Language for Non-unicode programs na Polski. 
 
W przypadku problemów czytania polskich znaków z Oracle (w SAS-ie na Windows) należy ustawić 
zmienną środowiskową NLS_LANG. Ta zmienna decyduje o tym w jakim encodingu narodowe znaki 
będą wysyłane i wczytywane przez klienta Oracle. Wartość zmiennej odpowiadającej koldowaniu 
wlatin2: 
NLS_LANG= POLISH_POLAND.EE8MSWIN1250. 
 
 
6. Zlokalizowane produkty. 
 

Dział NLS naszego biura zlokalizował lub lokalizuje następujące produkty: 
a)         SAS 9.3 
b)         AMO 4.3 
c)         BID 4.3 
d)         EG 4.3 
e)         EM 6.2 
f)          FCS 3.1 
g)         FM 5.2 
h)         FXA 5.2 
i)          IDP 4.3 
j)          ORM 4.3 
k)         SES 5.2 
l)          SMA 5.1 
m)        SM 5.2 
n)         WRS 4.3 
 
Rozszyfrowanie skrótów: 
AMO    Add-In for Microsoft Office 
BID   BI Dashboard 
EG       Enterprise Guide 
EM    Enterprise Miner 
FCS   Forecast Server 
FM       Financial Management Solutions 
FXA     Financial Management Add-In for Microsoft Office 
IDP      Information Delivery Portal 
ORM   OpRisk Monitor 
SES     Solution Services 
SMA    Solution Services Add-In for MS Office 
SM       Strategic Management 
WRS    Web Report Studio 
 

http://support.sas.com/kb/18/849.html

